

POLY 88 Software Library

Title Game Tape #1: LANDER, STOCK, REVERSE, HAMURABI

Hardware Requirements 16K RAM

Software Requirements BASIC version A00

by

**PolyMorphic
Systems**

Game Tape #1

FEATURES FOUR GAMES:

LANDER, a real challenge to land the lunar module on the platform.

STOCK, a chance to play the stock market without touching your bank account.

REVERSE, a puzzle to challenge kids of all ages.

HAMURABI, will Sumeria become rich under your rule?

These programs should be loaded after BASIC, version A00. Make sure the tape is rewound. Now "ZERO" your cassette recorder. Use the counter to locate approximate position of the program you are interested in.

<u>Program</u>		<u>Counter</u>	<u>Total Memory Required (less printer driver.)</u>
LANDER	Ø-17	000	16K RAM
STOCK	Ø-17	060	16K RAM
REVERSE		125	16K RAM
HAMURABI	Ø-17	177	16K RAM

The counter is based on a SUPERSCOPE 103-A. Only one program may be loaded at a time. Brief descriptions and listings for each program follow this page.

PROGRAM NAME: LANDER

TYPE: Game

HARDWARE REQUIREMENTS: 16K RAM

SOFTWARE REQUIREMENTS: BASIC Version A00

BRIEF DESCRIPTION:

You are in a lunar module falling toward the moon's surface. The onboard computer is not working and you must manually land, not crash, the module on a small platform. There is a set amount of fuel which can be used. No control is possible when the fuel is gone.

OPERATING INSTRUCTIONS:

Use keyboard commands (0 through 9) to increase or decrease the lander's rocket thrust. See the instructions within the program for more detail. You win if you contact the platform AND the comment reads "BY GEORGE, YOU'VE GOT IT!"

DESCRIPTION OF HOW PROGRAMS WORKS:

The lander has a constant horizontal velocity. In the vertical direction, the module is accelerated by the vector sum of gravity and rocket thrust. Each time the module changes position, the keyboard is sampled for the thrust value. Vertical velocity and fuel are adjusted accordingly. The new module position is then plotted on the video screen. A comment is given according to the contact velocity.


```
>>LIST100,600
100 REM *** LUNAR LANDER ***
110 DIM L$(60)
120 S= 63488
130 !CHR$(12),TAB(22),"LUNAR LANDER"
140 !\INPUT "WOULD YOU LIKE INSTRUCTIONS (Y OR N)?",I$
150 IF LEN(I$)=0 THEN 550
160 IF I$(1,1)<>"Y" THEN 550
170 !\!"YOU ARE THE PILOT OF A LUNAR LANDER MODULE, AND,"
180 !"UNFORTUNATELY, YOUR ON-BOARD COMPUTER BROKE DOWN"
190 !" (IT WAS NOT MADE BY POLYMORPHICS). YOU TOOK OVER"
200 !"MANUAL CONTROL TO LAND YOUR SHIP ON THE MOON."
210 !"INITIALLY, YOUR LANDER IS SEVERAL HUNDRED FEET"
220 !"ABOVE A LUNAR LANDING PLATFORM, AND FALLING TOWARD"
230 !"IT, ACCELERATING AS IT FALLS. YOUR OBJECTIVE IS TO"
240 !"MAKE A SOFT LANDING ON THE PLATFORM. YOUR ONLY"
250 !"AVAILABLE MANEUVER IS TO BRAKE WITH YOUR ROCKET"
260 !"ENGINE."!\!
270 GOSUB 1550
280 !CHR$(12),
290 !"1) ABOUT YOUR ENGINE: YOUR ENGINE HAS TWO LIMITATIONS:"
300 !" A) IT CAN ONLY EXERT A THRUST BETWEEN 0-9"
310 !" B) THEIR IS A LIMITED AMOUNT OF FUEL ON BOARD, YOU"
320 !" HAVE ENOUGH FOR A SAFETY FACTOR, BUT CONSERVE!"
330 !\!"2) UNITS OF FUEL: ONE UNIT OF FUEL SHALL REDUCE YOUR"
340 !" ACCELERATION BY ONE FOOT PER SECOND SQUARED, AND"
350 !" SINCE LUNAR ACCELERATION IS 2 FT/SECOND SQUARED,"
360 !" A BURN RATE OF 2 WILL HOLD YOUR SPEED CONSTANT."
370 !\!"3) THE LANDING PLATFORM: IT IS VERY IMPORTANT THAT"
380 !" YOU TRY TO LAND ON THE PLATFORM, OTHERWISE THE"
390 !" LANDER WILL NOT SURVIVE A LANDING ON THE MOON."!\!
400 GOSUB 1550 \ !CHR$(12),
410 !"REMEMBER - THE ONLY COMMANDS YOU CAN USE TO CONTROL"
420 !"YOUR SPEED ARE: 0 THROUGH 9 (BURN RATE). "
430 !"SMALLER BURN RATES MAKE THE LANDER FALL FASTER OR"
440 !"DECREASE UPWARD VELOCITY."
450 !"LARGER BURN RATES MAKE THE LANDER FALL SLOWER OR"
460 !"INCREASE UPWARD VELOCITY."
470 !"'2' KEEPS THE VELOCITY CONSTANT."!\!
480 !"THE LARGER THE BURN RATE, THE FASTER YOU USE FUEL."
490 !"IF YOU RUN OUT OF FUEL, YOU CAN'T CONTROL YOUR,"
500 !"SPEED - GRAVITY TAKES OVER!"
510 !\!"WHEN THE GAME DISPLAY APPEARS, YOUR LANDER WILL BE"
520 !"IN THE UPPER LEFT, PLATFORM ON THE BOTTOM."
530 !"AT THAT POINT, TYPE IN ANY BURN RATE TO BEGIN."
540 GOSUB 1550
550 !CHR$(12),
560 REM INITIALIZE DATA
570 T= 0
580 H= 500
590 V= 35+INT(10*RND(0))
600 F= 150+INT(50*RND(0))
>>
```



```
>>LIST 610,1100
610 B= 0
620 REM          INITIALIZE SCREEN
630 !CHR$(12),
640 REM THE STARS
650 FOR X= 0 TO 50
660 POKE S+INT(1020*RND(0)),174
670 NEXT X
680 REM THE LANDING PLATFORM
690 FOR X= 54 TO 74
700 PLOT X,0,1
710 NEXT X
720 GOSUB 1570 \ K=INP(1) \ GOTO 760
730 GOSUB 1570
740 REM          INPUT
750 IF INP(0)=0 THEN 760 ELSE K=INP(1)
760 B=K-48
770 IF B<0 OR B>9 THEN 750
780 REM          PROCESSING
790 T=T+1
800 IF T=127 THEN 1090
810 IF T=74 THEN 970
820 IF T>74 THEN B=2
830 IF B>F THEN B=F
840 F=F-B
850 V1= V-B/2+1
860 IF H<(V+V1)/2 THEN 1020
870 H= H-(V+V1)/2
880 IF H>500 THEN 1090
890 V= V1
900 IF F>0 THEN 730
910 REM          FREE-FALL ROUTINE
920 ! CHR$(11)
930 !" OUT OF FUEL - FREE FALL STARTS!"
940 B=0
950 GOSUB 1570
960 GOTO 790
970 ! CHR$(11)
980 !\!" YOU HAVE OVER SHOT THE LUNAR PLATFORM... BUCKLE YOUR"
990 !" SEATBELTS AND PREPARE FOR A CRASH LANDING!"
1000 GOTO 840
1010 REM CONTACT!
1020 IF T>54 THEN 1140
1030 REM          MISSED PLATFORM ROUTINE
1040 ! CHR$(11)
1050 !\!\!\!" YOU HAVE MISSED THE LANDING PLATFORM, AND"
1060 !" CRASHED ON THE LUNAR SURFACE, JUST BELOW."
1070 GOTO 1160
1080 REM          ORBIT ROUTINE
1090 !CHR$(11)\!\!\!\!\!"
1100 !" YOU HAVE OVER-CORRECTED YOUR BURN RATE TO COUNTERACT"
>>
```



```
>>LIST 1110,1600
1110 !" THE GRAVITATIONAL PULL AND WENT INTO ORBIT ..FOREVER!"
1120 GOTO 1520
1130 REM          CONACT WITH PLATFORM ROUTINE
1140 !CHR$(11)\!\!\!\!\!\!\!\!\!\!
1150 !" CONTACT WITH PLATFORM!"
1160 IF B=2 THEN D=H/V
1170 IF B=2 THEN 1190
1180 D= (-V+SQRT(V^2+H*(4-2*B))) / (2-B)
1190 V= V+D*(2-B)
1200 T= T+D
1210 H= 0
1220 GOSUB 1570
1230 REM          COMMENT ROUTINE
1240 !CHR$(11)
1250 RESTORE 1260
1260 DATA "BY GEORGE, YOU'VE GOT IT!"
1270 DATA "WHEN YOUR WHIPLASH SUBSIDES, HAVE ANOTHER GO!"
1280 DATA "WELL, BACK TO THE OLD DRAWING BOARD."
1290 DATA "DID YOU KNOW THAT THIS LEM COSTS $114,942,124.99?"
1300 DATA "PLEASE DON'T LITTER."
1310 DATA "WITH A POLY-88 TO HELP YOU, YOU MIGHT OF MADE IT!"
1320 DATA "SIMPLY SMASHING!"
1330 DATA "TO DIE, TO SLEEP, NO MORE....."
1340 DATA "IT'S A BIRD! IT'S A PLANE! IT'S A MISCALCULATION!"
1350 DATA "ACTION UNGOVERNED BY THOUGHT ENDS IN DIASTER"
1360 DATA "WELCOME TO NEGATIVE POPULATION GROWTH!"
1370 DATA "LIKE A BRICK!"
1380 DATA "CLOSE DOESN'T COUNT"
1390 DATA "AH! TO HAVE TO COME SO CLOSE!"
1400 DATA "OOH! ..... I CAN'T LOOK AT THE MESS!"
1410 DATA "BLOOD, GUTS, TWISTED METAL, YUCK!"
1420 T1= INT(V/3.1)
1430 IF T1>15 THEN T1=15
1440 FOR X= 0 TO T1
1450 READ L$
1460 NEXT X
1470 !CHR$(11)\!\!\!\!\!\!\!\!\!\!
1480 !" ",L$
1490 IF V*.27 < 3 THEN 1520
1500 !\!" AND YOU HAVE JUST BLASTED A NEW LUNAR CRATER",V*.27,
1510 !" FEET DEEP!"
1520 !CHR$(16+8) \ GOSUB 1550
1530 GOTO 560
1540 REM
1550 INPUT"TYPE RETURN KEY NOW TO CONTINUE....",Z$\RETURN
1560 REM          HEADER PRINT-OUT SUBROUTINE
1570 !%#Z6F1,
1580 !CHR$(11),
1590 !" TIME=",T,
1600 !" HEIGHT=",H,
>>
```



```
>>LIST 1610
1610 !" VELOCITY=",V,
1620 !" FUEL=",F,
1630 !" BURN=",%2I,B
1640 REM          PLOT LEM ROUTINE
1650 IF H/10>46 THEN 1680
1660 PLOT T,H/10,1
1670 PLOT 0,0,0
1680 RETURN
1690 REM EXIT ROUTINE
1700 !CHR$(12),
1710 FOR I=1 TO 10
1720 K=INP(1)
1730 !K,K-48
1740 NEXT
1750 !CHR$(12)
>>
```


PROGRAM NAME: STOCK

TYPE: Game

HARDWARE REQUIREMENTS: 16K RAM

SOFTWARE REQUIREMENTS: BASIC Version A00

BRIEF DESCRIPTION:

Now you can play the stock market without financial worry. The computer simulates the rise and fall trends of the market.

OPERATING INSTRUCTIONS:

Use the keyboard to indicate the number of shares you wish to buy (positive number) or sell (negative number).

DESCRIPTION OF HOW PROGRAM WORKS:

A combination of random factors and trends are combined to change the rise and fall of stock prices. The market has an overall trend as well as individual trends for each stock.


```
>>LIST100,600      *** STOCK ***
100 REM
110 REM Edited by T.Jensen from public domain
120 !CHR$(12),TAB(20),"THE STOCK MARKET"
130 DIM S(5),P(5),V(5),Z(5),C(5)
150 Z=RND(TIME(1)/65535)
160 A=FNA(RND(0)/10)
170 T5=0\X9=0\N1=0\N2=0\E1=0\E2=0
180 INPUT "Would you like instructions? (Y or N)",I$
190 !CHR$(12)
200 IF I$="N" THEN 400
210 IF I$="n" THEN 400
220 !"This program plays the stock market. You will be given"
230 !" $10,000 and may buy or sell stocks. The stock prices"
240 !" will be generated randomly and therefore this model"
250 !" does not represent exactly what happens on the"
260 !" exchange."!\!\!"If you wish to stop playing at any ",
265 !" time, oversell a stock." \!\GOSUB 1340
270 !" A table of available stocks, their prices, and the"
275 !" number of shares in your portfolio will be printed."
280 !" Following this, the initials of each stock will be"
290 !" printed with a question mark. Here you must decide"
300 !" how many shares you wish to buy or sell. A positive"
310 !" number indicates you're buying, a negative number"
320 !" indicates you're selling (0=no change). A brokerage"
330 !" fee will be charged on all transactions. Note that if"
340 !" a stock's value drops to zero, it can rebound to a"
350 !" positive value. You have $10,000 to invest. Use integers"
360 !" for all your inputs. (NOTE: to get a 'feel' for the"
370 !" market, try running a few days without buying)"
380 !" -----GOOD LUCK-----"
390 GOSUB 1340
400 !"Working...."
430 S(1)=100\S(2)=85\S(3)=150\S(4)=140\S(5)=110
440 N1$="IBM"\N2$="RCA"\N3$="LBJ"\N4$="ABC"\N5$="CBS"
460 T8=FNB(0)
480 IF RND(0)<.5 THEN A=-A
500 GOSUB 1360
520 FOR I=1 TO 5\P(I)=0\Z(I)=0\NEXT
530 !\!
550 C=10000
570 !CHR$(12),
580 !TAB(9),"STOCK",TAB(31),"INITIALS PRICE/SHARE"
590 !"Int. Ballistic Missiles",TAB(33),"IBM",%#15F2,S(1)
600 !"Red Cross of America",TAB(33),"RCA",S(2)
>>
```



```
>>LIST 610,1100
610 !"Lichtenstein, Bumrap & Joke",TAB(33),"LBJ",S(3)
620 !"American Bankrupt Company",TAB(33),"ABC",S(4)
630 !"Censured Books Store",TAB(33),"CBS",S(5)
640 !
660 Z4=Z5\Z5=0\T=0
670 FOR I =1 TO 5
680 Z5=Z5+S(I) \ T=T+(S(I)*P(I))
690 NEXT
700 Z5=FNA(Z5/5)
710 Z6=FNA(Z5-Z4)
730 D=T+C
740 !"New York Stock Exchange AVERAGE:",%#9F2,Z5,
750 IF X9=0 THEN ! ELSE !" Net Change:",Z6,
760 !
770 T=FNA(T) \ C=FNA(C) \ D=FNA(D)
780 !"Total STOCK ASSETS are: ",%#Scl1F2,T
790 !"Total CASH ASSETS are:",TAB(24),C
800 !"Total ASSETS are:",TAB(24),D \!
810 IF X9=0 THEN 840
820 Z=RND(TIME(1)/65535)
840 !"What is your transaction in:"
850 INPUT "IBM? ",Z(1) \ INPUT "RCA? ",Z(2)
860 INPUT "LBJ? ",Z(3) \ INPUT "ABC? ",Z(4)
870 INPUT "CBS? ",Z(5) \!
890 P5=0 \ S5=0
900 FOR I=1 TO 5
910 Z(I)=INT(Z(I)+.5)
920 IF Z(I)<=0 THEN 950
930 P5=P5+(Z(I)*S(I))
940 GOTO 1020
950 S5=S5-Z(I)*S(I)
960 IF -Z(I)<=P(I) THEN 1020
970 !"You have oversold a stock."
980 !"You can now opt to stop playing (type S)"
985 INPUT "or continue (C)",IS
990 IF IS="S" THEN 1820
1000 IF IS="s" THEN 1820
1010 !CHR$(12),\GOTO 1230
1020 NEXT I
1040 T5=P5+S5
1060 B5=FNA(.01*T5)
1090 C5=C-P5-B5+S5
1100 IF C5>=0 THEN 1130
>>
```



```
>>LIST1110,1600
1110 !"You have used",%$C13F2,-C5," more then you have."
1120 !\GOSUB 1340 \ !CHR$(12), \ GOTO 1230
1130 LETC=C5
1150 FOR I=1 TO 5
1160 P(I)=P(I)+Z(I)
1170 NEXT I
1190 GOSUB 1360
1200 FOR I=1 TO 5 \ V(I)=P(I)*S(I) \ NEXT
1220 !CHR$(12),"*** End of day's trading ***" \ !
1230 !"STOCK PRICE/SHARE HOLDINGS VALUE NET ",
1240 !"PRICE CHANGE"
1250 FOR I=1 TO 5 \ IF I=1 THEN !N1$,
1260 IF I=2 THEN !N2$,
1270 IF I=3 THEN !N3$,
1280 IF I=4 THEN !N4$, ELSE IF I=5 THEN !N5$,
1290 !%#11F2,TAB(6),S(I),TAB(25),%,P(I),
1300 IF V(I)=0 THEN !TAB(36),%,0, ELSE !TAB(33),V(I),
1310 !TAB(44),C(I) \ NEXT
1320 X9=1
1330 GOTO 640
1340 INPUT "Type RETURN key now to continue...",I$
1350 !CHR$(12)\RETURN
1360 IF N1>0 THEN 1400
1370 I1=FNB(0) \ N1=FNB(0)
1380 E1=1
1400 IF N2>0 THEN 1450
1410 I2=FNB(0) \ N2=FNB(0)
1420 N2=INT(4.99*RND(X)+1)
1430 E2=1
1450 N1=N2-L
1460 N2=N2-1
1480 FOR I=1 TO 5
1490 X1=RND(X)
1500 IF X1<=.25 THEN X1=.25\GOTO 1540
1510 IF X1<=.50 THEN X1=.50\GOTO1540
1520 IF X1<=.75 THEN X1=.75 ELSE X1=0
1540 W3=0
1550 IF E1<1 THEN 1600
1560 IF INT(I1+.5)=(I+.5) THEN W3=10\E1=0
1580 W3=10
1590 E1=0
1600 IF E2<1 THEN 1660
>>
```



```
>>LIST 1610
1610 IF INT(12+.5)<>INT(I+.5) THEN 1660
1630 W3=W3-10
1640 E2=0
1660 C(I)=INT(A*S(I))+X1+INT(3-(6*RND(0))+.5)+W3
1670 C(I)=FNA(C(I))
1680 S(I)=S(I)+C(I)
1690 IF S(I)<0 THEN C(I)=0\S(I)=0\GOTO 1710
1700 S(I)=FNA(S(I))
1710 NEXT I
1730 T8=T8-1
1740 IF T8>=1 THEN RETURN
1750 T8=FNB(0)
1760 A=FNA(RND(0)/10)
1770 S4=RND(0)
1780 IF S4>.5 THEN RETURN ELSE A=-A
1790 RETURN
1800 DEF FNA(X)=INT(100*X+.5)/100
1810 DEF FNB(X)=INT(4.99*RND(TIME(X)/65535)+1)
1820 !CHR$(12)\!\!\! TAB(20),%#,"Hope you had fun!"
>>
```


PROGRAM NAME: REVERSE

TYPE: Game

HARDWARE REQUIREMENTS: 16K RAM

SOFTWARE REQUIREMENTS: BASIC Version A00

BRIEF DESCRIPTION:

Reverse is a mathematical puzzle. The object is to sequentially arrange a random list of numbers. The user can only reverse blocks of numbers.

OPERATING INSTRUCTIONS:

The keyboard is used to input how many numbers the user wants to reverse. Numbers must be reversed in blocks and all blocks must begin with the left-most number.

DESCRIPTION OF HOW PROGRAM WORKS:

The computer creates a random list of numbers and performs the busy work of reversing blocks of numbers.


```
>LIST100,600
100 REM *** REVERSE ***
110 DIM A(30),A$(16)
120 A$=" " \REM 16 SPACES
130 PRINT CHR$(12),TAB(25),"REVERSE"
140 T=RND(TIME(1)/65535)
150 INPUT "Would you like instructions? (Y or N)",I$
160 IF I$="N" THEN 330
170 IF I$="n" THEN 330
180 !CHR$(12),"This is the game of REVERSE. To win, all you"
190 !"have to do is arrange an unsorted, sequential"
200 !"list of integers. The final arrangement must"
210 !"have 0 on the left and 'count up' to the highest"
220 !"integer (on the right). You can only reverse"
230 !"the order of a block of integers. This block"
240 !"can be any length but must begin with the left-"
250 !"most integer. For example, if the current list is:"
260 !"1 2 3 4 5 0 6 7 8 9"
270 !"and you reverse 5, you will then have:"
280 !"5 4 3 2 1 0 6 7 8 9"
290 !"Note the positions of the first 5 and last 5 digits."
300 !"Now, if you reverse 6, you will win:"
310 !"0 1 2 3 4 5 6 7 8 9"
320 !"If you want to change the array size, reverse 0!"
330 REM CLE THE INPUT BUFFER FLAG.
340 INPUT "How many digits do you want to arrange?",M
350 IF M<=0 THEN 1080
360 IF M<=17 THEN 410
370 !"I'm sorry, but I can't go beyond 17 array elements."
380 GOTO 330
390 G=0
400 REM MAIN LOOP
410 G= G+1
420 X= 0
430 !CHR$(12),"Game #",G
440 !"Thinking.....",
450 REM BUILD A LIST
460 FOR L1=0 TO M-1
470 A(L1)=INT (M*(RND(TIME(1)/65535)))
480 IF L1=0 THEN 540
490 REM CHECK FOR REPEATS
500 L2= 0
510 IF A(L1)=A(L2) THEN 470
520 L2=L2+1
530 IF L2<L1 THEN 510
540 NEXT L1
550 REM DISPLAY A LIST
560 !CHR$(12)," POSITIONS:",
570 FOR L1=1 TO M
580 !%3I,L1,
590 NEXT L1
600 !CHR$(11) \!" THE LIST: ",
>LIST610,
```



```
610 FOR L1=0 TO M-1
620 !%3I,A(L1),
630 NEXT L1
640 IF W=1 THEN RETURN
650 X= X+1
660 !\!"This is your try #",X
670 !"How many integers do you want to reverse?",
680 REM A$ + 6 SPACES
690 !A$,"      ",\POKE 63743,160
700 PLOT 80,37,0\INPUT R$
710 PLOT 0,34,0 \FOR I=1 TO 6 \!A$, \NEXT
720 POKE 63840,160 \PLOT 0,34,0
730 IF LEN(R$)<3 THEN 760
740 !"I can't use a number with more than 2 digits,"
750 !"even if it's correct!" \GOTO 600
760 IF LEN(R$)=0 THEN 810
770 FOR I=1 TO LEN(R$)
780 R1=ASC(R$(I,I))
790 IF (R1<48) OR (R1>57) THEN 810
800 NEXT \R=VAL(R$) \GOTO 820
810 !"Input error - retype!" \GOTO 600
820 IF R<0 THEN !"That's a strange choice--try again!"\GOTO 600
830 IF R=0 THEN 330
840 IF R<=M THEN 870
850 !"Please remember, you can't reverse more than",M
860 GOTO 600
870 REM REVERSE R NUMBERS IN LIST A
880 FOR L1= 0 TO INT(R/2)-1
890 T= A(L1)
900 A(L1)= A(R-L1-1)
910 A(R-L1-1)= T
920 NEXT L1
930 REM CHECK FOR A WIN
940 FOR L1=0 TO M-1
950 IF A(L1)<>L1 THEN !CHR$(12), \GOTO 560
960 NEXT L1
970 W=1 \GOSUB 560
980 !\!\!
990 !"And you WON in",X," move", \IF X=1 THEN !"!" ELSE !"s!"
1000 !\W=0
1010 INPUT "Type return key when you are ready to continue",I$
1020 IF G<10 THEN 410
1030 INPUT "You've played many games, are you getting bored?",I$
1040 IF I$="y" THEN 1080
1050 IF I$="Y" THEN 1080
1060 G=5 \GOTO 410
1070 REM EXIT ROUTINE
1080 !CHR$(12),"BYE!"
>
```


PROGRAM NAME: HAMURABI

TYPE: Game

HARDWARE REQUIREMENTS: 16K RAM

SOFTWARE REQUIREMENTS: BASIC Version A00

BRIEF DESCRIPTION:

Rule ancient Sumeria for ten years and reap the praise of an admiring populace or scorn of dissatisfied kingdom. The object is to make decisions that lead to bountiful harvest and few starvations.

OPERATING INSTRUCTIONS:

Your steward will provide statistics and ask for decisions concerning domestic problems. You will input your decisions via the keyboard.

DESCRIPTION OF HOW PROGRAM WORKS:

Keyboard inputs and a few random factors combine to determine the effects of each year.


```
>LIST 100,600
100 REM *** HAMURABI, ECONOMIC STRATEGY GAME, 22 FEB 77
110 REM CONVERTED FROM THE EDUSYSTEM 70 AND MODIFIED
120 REM FOR POLY 88 BY CINDY FEENEY, POLYMORPHIC SYSTEMS
130 !CHR$(12)
140 ! "HAMURABI: DO YOU WISH TO VISIT THE ORACLE TO HEAR"
150 ! "REVELATIONS CONCERNING THE WISE AND PROFITABLE RULE"
160 INPUT "OF ANCIENT SUMERIA DURING YOUR TEN YEAR TERM? (Y OR N)",A$
170 IF A$(1)="Y" THEN GOSUB 1390
180 B=RND(TIME(0)/65536)\D1=0\P1=0
190 Z=0\P=95\S=2800\H=3000\E=H-S
200 Y=3\A=H/Y\I=5\Q=1
210 D=0
220 !CHR$(12)
230 PRINT\PRINT\PRINT "HAMURABI: I BEG TO REPORT TO YOU","\Z=Z+1
240 PRINT "IN YEAR",Z,"","D," PEOPLE STARVED","I," CAME TO THE CITY."
250 P=P+I
260 IF Q>=0 THEN 290
270 P=INT(P/2)
280 PRINT "A HORRIBLE PLAGUE STRUCK! HALF THE PEOPLE DIED."
290 PRINT "POPULATION IS NOW",P,"."
300 PRINT "THE CITY NOW OWNS",A," ACRES."
310 PRINT "YOU HARVESTED",Y," BUSHEL PER ACRE."
320 PRINT "RATS ATE",E," BUSHEL."
330 PRINT "YOU NOW HAVE",S," BUSHEL IN STORE."!\
340 IF Z=11 THEN GOTO 1120
350 C=INT(10*RND(0))\Y=C+17
360 PRINT "LAND IS TRADING AT",Y," BUSHEL PER ACRE."
370 PRINT "HOW MANY ACRES DO YOU WISH TO BUY",
380 INPUT Q\IF Q<0 THEN 1090
390 IF Y*Q<=S THEN 420
400 GOSUB 1010
410 GOTO 370
420 IF Q=0 THEN 450
430 LET A=A+Q\S=S-Y*Q\C=0
440 GOTO 510
450 PRINT "HOW MANY ACRES DO YOU WISH TO SELL",
460 INPUT Q\IF Q<0 THEN 1090
470 IF Q<A THEN 500
480 GOSUB 1040
490 GOTO 450
500 A=A-Q\S=S+Y*Q\C=0
510 PRINT
520 PRINT "HOW MANY BUSHEL DO YOU WISH TO FEED YOUR PEOPLE",
530 INPUT Q
540 IF Q<0 THEN 1090
550 REM TRYING TO USE MORE GRAIN THAN IN THE SILOS?
560 IF Q<=S THEN 590
570 GOSUB 1010
580 GOTO 520
590 S=S-Q\C=1\PRINT
600 PRINT "HOW MANY ACRES DO YOU WISH TO PLANT WITH SEED",
>
```



```
>LIST 610,1100
610 INPUT D\IF D=0 THEN 750
620 IF D<0 THEN 1090
630 REM TRYING TO PLANT MORE ACRES THAN YOU OWN?
640 IF D<=A THEN 670
650 GOSUB 1040 \GOTO 600
660 REM ENOUGH GRAIN FOR SEED?
670 IF INT(D/2)<S THEN 710
680 GOSUB 1010
690 GOTO 600
700 REM ENOUGH PEOPLE TO TEND THE CROPS?
710 IF D<10*P THEN 740
720 !"BUT YOU HAVE ONLY",P," PEOPLE TO TEND THE FIELDS. NOW THEN,"
730 GOTO 600
740 S=S-INT(D/2)
750 GOSUB 1070
760 REM A BOUNTIFUL HARVEST!
770 Y=C\H=D*Y\ E=0
780 GOSUB 1070
790 IF INT(C/2)<>C/2 THEN 820
800 REM THE RATS ARE RUNNING WILD!
810 E=INT(S/C)
820 S=S-E+H
830 GOSUB 1070
840 REM LET'S HAVE SOME BABIES.
850 I=INT(C*(20*A+S)/P/100+1)
860 REM HOW MANY PEOPLE HAD FULL TUMMIES.
870 C=INT(Q/20)
880 REM HORRORS, A 15% CHANCE OF PLAGUE.
890 Q=INT(10*(2*RND(0)-.3))
900 IF P<C THEN 210
910 REM STARVE ENOUGH FOR IMPEACHMENT?
920 D=P-C\IF D>.45*P THEN 950
930 P1=((Z-1)*P1+D*100/P)/Z
940 P=C\D1=D1+D\GOTO 230
950 !\! "YOU STARVED",D," PEOPLE IN ONE YEAR!!!"
960 ! "WHAT DID THE ORACLE TELL YOU?"
970 PRINT "DUE TO THIS EXTREME MISMANAGEMENT YOU HAVE NOT ONLY"
980 PRINT "BEEN IMPEACHED AND THROWN OUT OF OFFICE BUT YOU "
990 PRINT "HAVE ALSO BEEN EXILED TO PERSIA!"
1000 GOTO 1370
1010 PRINT "HAMURABI: THINK AGAIN. YOU HAVE ONLY"
1020 PRINT S," BUSHEL OF GRAIN. NOW THEN,"
1030 RETURN
1040 PRINT "HAMURABI: THINK AGAIN. YOU OWN ONLY",A," ACRES."
1050 PRINT "NOW THEN,"
1060 RETURN
1070 C=INT(RND(0)*5)+1
1080 RETURN
1090 PRINT\PRINT "HAMURABI: I CANNOT DO WHAT YOU WISH."
1100 PRINT "GET YOURSELF ANOTHER STEWARD!!!"
```



```
>LIST 1110
1110 GOTO 1370
1120 ! "IN YOUR TEN YEAR TERM OF OFFICE","P1," PERCENT OF"
1130 PRINT "THE POPULATION STARVED PER YEAR ON THE AVERAGE."
1140 PRINT "A TOTAL OF",D1," PEOPLE DIED!!"\L=A/P
1150 PRINT "YOU STARTED WITH TEN ACRES PER PERSON AND ENDED"
1160 PRINT "WITH",L," ACRES PER PERSON."!\
1170 INPUT"DO YOU WISH TO HEAR THE ORACLE'S EVALUATION?",A$
1180 IF A$(1)<>"Y" THEN GOTO 1370
1190 !CHR$(12)
1200 IF P1>33 THEN 970
1210 IF L<7 THEN 970
1220 IF P1>10 THEN 1300
1230 IF L<9 THEN 1300
1240 IF P1>3 THEN 1340
1250 IF L<10 THEN 1340
1260 PRINT "A FANTASTIC PERFORMANCE! CHARLEMANE, DISRAELI,"
1270 ! "AND JEFFERSON COMBINED COULD NOT HAVE DONE BETTER!"
1280 ! "THE ORACLE OF SUMERIA IS PLEASED."
1290 GOTO 1370
1300 PRINT "YOUR HEAVY-HANDED PERFORMANCE SMACKS OF NERO AND"
1310 PRINT "IVAN IV. THE PEOPLE (REMAINING) FIND YOU AN"
1320 PRINT "UNPLEASANT RULER, AND, FRANKLY, HATE YOUR GUTS!"
1330 GOTO 1370
1340 PRINT "YOUR PERFORMANCE COULD HAVE BEEN SOMEWHAT BETTER,"
1350 ! "BUT REALLY WASN'T TOO BAD AT ALL.",INT(P*.8*RND(0))
1360 ! "PEOPLE WOULD DEARLY LIKE TO SEE YOU ASSASSINATED."
1370 PRINT "SO LONG -- BACK TO BASIC."
1380 !\!\!\GOTO 1510
1390 ! CHR$(12)
1400 ! "YOUR STEWARD IS WISE AND DEVOTED TO BRING YOU TO VISIT."
1410 ! "HEED MY ADVICE WELL, HAMURABI."
1420 !\!\!\! "TO AVOID IMMEDIATE EXILE, DO NOT STARVE MORE THAN"
1430 !"HALF THE PEOPLE DURING ANY SINGLE YEAR."
1440 !\! "TO AVOID EXILE AFTER TEN YEARS, KEEP THE AVERAGE RATIO OF"
1450 !"ACRES TO PEOPLE AT LEAST SEVEN TO ONE."
1460 !\!
1470 INPUT"DO YOU WISH TO BEGIN YOUR FIRST YEAR? (Y OR N)",A$
1480 IF A$="Y" THEN RETURN
1490 GOTO 1370
1500 STOP
1510 PRINT ""
>
```